


---

Rolf Müller

# Mechanische Systeme mit Maple analysieren – Torsionsschwingungen berechnen


Zusatzmaterial zu



## Torsionsschwingungen mit Übertragungsmatrizen berechnen


### Grundlagen

Im Abschnitt 5.3.5 wurde das Modell eines rotierenden Zweimassensystems entwickelt und aus der Lösung des Differentialgleichungssystems u. a. eine Formel für die Eigenkreisfrequenz des Zweimassensystems abgeleitet. Oft interessiert aber das Schwingungsverhalten eines rotierenden Systems lediglich im stationären, eingeschwungenen Zustand. Auch haben die Systeme nicht selten mehr als zwei Drehmassen. In einem solchen Fall ist das im Folgenden dargestellte Verfahren für die Berechnung der Eigenkreisschwingungen gut geeignet. Es wird am Beispiel des im Bild 5.11 gezeigten Rotationssystems, bestehend aus drei Drehmassen mit den Trägheitsmomenten  $J_i$ , die durch zwei als masselos angenommene Wellen mit den Federkonstanten  $c_i$  verbunden sind, beschrieben.



**Bild 0.1** Rotierendes System mit drei Massen

Das Verfahren beruht auf der Aufteilung des Gesamtsystems in Segmente (oft auch Felder genannt). Die Schnittstellen zwischen den Segmenten werden durch die dort wirkenden Zustandsgrößen, im vorliegenden Beispiel Drehwinkel  $\varphi$  und Moment  $M$ , beschrieben. Den Zusammenhang zwischen den Zustandsgrößen am rechten und am linken Rand eines Segments liefert eine Matrix (Übertragungsmatrix), die für jedes Segment aus Gleichgewichtsbeziehungen und kinematischen Zwangsbedingungen bestimmt werden kann. Im Bild 5.12 ist das segmentierte System mit den Zustandsgrößen dargestellt. Die Indizes der Trägheitsmomente und der Federkonstanten sind darin der Bezeichnung der Segmente  $k$  angepasst.



**Bild 0.2** Schnittstellen und Segmente des Rotationssystems

Einführend werden die Grundlagen des Verfahrens beschrieben [Göld67, HoDre94]. Für die Drehmasse (Scheibe) des Segments  $k$  im Bild 5.12 gilt

$$\varphi_k = \varphi_{k-1} \quad (5.51)$$

Der Drehwinkel  $\varphi$  wird in der Wirkungsrichtung des Momentes  $M$  auf der linken Seite einer Scheibe als positiv angenommen. Damit ist

$$M_{k-1} - M_k - J_k \cdot \ddot{\varphi} = 0 \quad (5.52)$$

Für die ungedämpfte Eigenschwingung ist bekanntlich

$$\varphi = A \cdot e^{j\omega t}; \quad \ddot{\varphi} = -A \cdot \omega^2 \cdot e^{j\omega t} = -\omega^2 \cdot \varphi \quad (5.53)$$

Gl. (5.53) und (5.51) in (5.52) eingesetzt liefert

$$M_k = M_{k-1} + J_k \cdot \omega^2 \cdot \varphi_{k-1} \quad (5.54)$$

Durch (5.51) und (5.54) werden damit die Zustandsgrößen rechts von der Drehmasse im Segment  $k$  durch die Zustandsgrößen links von ihr ausgedrückt. In Matrizenschreibweise ergibt sich

$$\begin{pmatrix} \varphi_k \\ M_k \end{pmatrix} = \begin{pmatrix} 1 & 0 \\ J_k \cdot \omega^2 & 1 \end{pmatrix} \cdot \begin{pmatrix} \varphi_{k-1} \\ M_{k-1} \end{pmatrix} \quad (5.55)$$

bzw. in Kurzschreibweise

$$\mathbf{z}_k = \mathbf{T}_{\mathbf{M},k} \cdot \mathbf{z}_{k-1} \quad \text{mit} \quad \mathbf{z}_k = \begin{pmatrix} \varphi_k \\ M_k \end{pmatrix} \quad \mathbf{T}_{\mathbf{M},k} = \begin{pmatrix} 1 & 0 \\ J_k \cdot \omega^2 & 1 \end{pmatrix} \quad (5.56)$$

Die Matrix  $\mathbf{T}_{\mathbf{M},k}$  wird als Übertragungsmatrix der Drehmasse im Segment  $k$  bezeichnet. Die Vektoren  $\mathbf{z}_k$  und  $\mathbf{z}_{k-1}$  sind die Zustandsvektoren der rechten und der linken Seite der Drehmasse oder allgemeiner des Segments  $k$  (Bild 5.12).

Die Beziehung zwischen den Zustandsgrößen auf der linken und der rechten Seite einer Torsionsfeder (Welle) mit der Federkonstanten  $c_i$  lässt sich ebenfalls durch eine Matrizengleichung beschreiben. Für jede Welle des Segments  $k$  gilt das Momentengleichgewicht

$$M_k = M_{k-1} \quad (5.57)$$

und

$$M_{k-1} = c_k (\varphi_{k-1} - \varphi_k) \quad (5.58)$$

$$\varphi_k = \varphi_{k-1} - \frac{M_{k-1}}{c_k} \quad (5.59)$$

Aus (5.57) und (5.59) folgt daher für die Torsionsfeder

$$\begin{pmatrix} \varphi_k \\ M_k \end{pmatrix} = \begin{pmatrix} 1 & -1/c_k \\ 0 & 1 \end{pmatrix} \cdot \begin{pmatrix} \varphi_{k-1} \\ M_{k-1} \end{pmatrix} \quad (5.60)$$

bzw. in Kurzschreibweise

$$\mathbf{z}_k = \mathbf{T}_{F,k} \cdot \mathbf{z}_{k-1} \quad \text{mit} \quad (5.61)$$

$$\mathbf{z}_k = \begin{pmatrix} \varphi_k \\ M_k \end{pmatrix} \quad \mathbf{T}_{F,k} = \begin{pmatrix} 1 & -1/c_k \\ 0 & 1 \end{pmatrix}$$

Ein Schwingungssystem gemäß Bild 5.12 kann man nun stufenweise wie folgt berechnen:

$$\mathbf{z}_0 \rightarrow \mathbf{z}_1 := \mathbf{T}_1 \cdot \mathbf{z}_0 \rightarrow \mathbf{z}_2 := \mathbf{T}_2 \cdot \mathbf{z}_1 \rightarrow \mathbf{z}_3 := \mathbf{T}_3 \cdot \mathbf{z}_2 \rightarrow \mathbf{z}_4 := \mathbf{T}_4 \cdot \mathbf{z}_3 \rightarrow \mathbf{z}_5 := \mathbf{T}_5 \cdot \mathbf{z}_4$$

Die Randbedingungen  $\mathbf{z}_0$  bzw.  $\mathbf{z}_5$  sind durch die konkreten Bedingungen vorgegeben. Für das im Bild 5.12 gezeigte Beispiel sind die Elemente des Zustandsvektors  $\mathbf{z}_0$  folgendermaßen definiert:  $M_0 = 0$ ;  $\varphi_0$  unbekannt.

## Berechnungsbeispiel

Die Umsetzung des beschriebenen Verfahrens in Maple wird anhand des in Bild 5.12 dargestellten Rotationssystems gezeigt.

```
> interface(displayprecision=3):
```

**Notierung der Übertragungsmatrizen  $T_k$ :**

```
> T[1] := Matrix([[1,0],[J[1]*omega^2, 1]]);
```

$$T_1 := \begin{bmatrix} 1 & 0 \\ J_1 \omega^2 & 1 \end{bmatrix}$$

```
> T[2] := Matrix([[1, -1/c[1]],[0, 1]]);
```

$$T_2 := \begin{bmatrix} 1 & -\frac{1}{c_1} \\ 0 & 1 \end{bmatrix}$$

```
> T[3] := Matrix([[1,0],[J[2]*omega^2, 1]]):
```

```
> T[4] := Matrix([[1, -1/c[2]],[0, 1]]):
```

```
> T[5] := Matrix([[1,0],[J[3]*omega^2, 1]]):
```

**Vorgabe der Bedingungen am linken Rand:**

```
> M[0] := 0:
```

```
> z[0] := Vector([phi[0], M[0]]):
```

**Berechnung der Zustandsvektoren**

In der for-Schleife werden die Vektoren aller Zustandsgrößen des Schwingungssystems ermittelt.

```
> for k from 1 to 5 do
    z[k] := T[k].z[k-1];
end do;
```

$$\begin{aligned}
z_1 &:= \begin{bmatrix} \phi_0 \\ J_1 \omega^2 \phi_0 \end{bmatrix} \\
z_2 &:= \begin{bmatrix} \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \\ J_1 \omega^2 \phi_0 \end{bmatrix} \\
z_3 &:= \begin{bmatrix} \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \\ J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0 \end{bmatrix} \\
z_4 &:= \begin{bmatrix} \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} - \frac{J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0}{c_2} \\ J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0 \end{bmatrix} \\
z_5 &:= \left[ \begin{bmatrix} \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} - \frac{J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0}{c_2} \\ J_3 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} - \frac{J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0}{c_2} \right) + J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0 \end{bmatrix} \right]
\end{aligned}$$

Das Moment am rechten Rand des Schwingungssystems wird durch die 2. Komponente des zuletzt berechneten Zustandsvektors  $\mathbf{z}_5 = \mathbf{z}[5]$  beschrieben. Der folgende Befehl stellt dieses nach Potenzen von  $\omega$  geordnet dar.

```
> collect(z[5][2], omega);
```

$$\frac{J_3 J_2 J_1 \phi_0 \omega^6}{c_2 c_1} + \left( -\frac{J_2 J_1 \phi_0}{c_1} + J_3 \left( -\frac{J_1 \phi_0}{c_1} - \frac{J_2 \phi_0 + J_1 \phi_0}{c_2} \right) \right) \omega^4 + (J_3 \phi_0 + J_1 \phi_0 + J_2 \phi_0) \omega^2$$

### Berechnung der Eigenkreisfrequenzen

Weil die rechte Seite der Drehmasse 3 frei ist, muss das Moment  $M$  an dieser Stelle Null sein. Unter Verwendung der Bedingung  $z[5][2] = 0$  können nun die Eigenkreisfrequenzen berechnet werden. Um zu einigermaßen übersichtlichen Ausdrücken zu gelangen, werden zuerst die Quadrate der Eigenkreisfrequenzen bestimmt. Dazu wird  $\omega^2$  in  $zr[5][2]$  durch die Variable  $om2$  ersetzt und dann die neue Gleichung  $w$  nach dieser Variablen aufgelöst.

```
> w:= subs (omega^2=om2, z[5][2]) :
```

Quadrate der Eigenkreisfrequenzen:

```
> EWq:= solve(w, om2) ;
```

$$\begin{aligned} EWq := 0, & \frac{1}{2} \frac{1}{J_3 J_2 J_1} \left( J_3 J_2 c_1 + J_3 J_1 c_1 + J_3 J_1 c_2 + J_2 c_2 J_1 \right. \\ & + \left( J_3^2 J_2^2 c_1^2 + 2 J_3^2 J_2 c_1^2 J_1 - 2 J_3^2 J_2 c_1 J_1 c_2 - 2 J_3 J_2^2 c_1 c_2 J_1 + J_3^2 J_1^2 c_1^2 \right. \\ & + \left. 2 J_3^2 J_1^2 c_1 c_2 - 2 J_3 J_1^2 c_1 J_2 c_2 + J_3^2 J_1^2 c_2^2 + 2 J_3 J_1^2 c_2^2 J_2 + J_2^2 c_2^2 J_1^2 \right)^{1/2}, \\ & - \frac{1}{2} \frac{1}{J_3 J_2 J_1} \left( -J_3 J_2 c_1 - J_3 J_1 c_1 - J_3 J_1 c_2 - J_2 c_2 J_1 \right. \\ & + \left( J_3^2 J_2^2 c_1^2 + 2 J_3^2 J_2 c_1^2 J_1 - 2 J_3^2 J_2 c_1 J_1 c_2 - 2 J_3 J_2^2 c_1 c_2 J_1 + J_3^2 J_1^2 c_1^2 \right. \\ & + \left. 2 J_3^2 J_1^2 c_1 c_2 - 2 J_3 J_1^2 c_1 J_2 c_2 + J_3^2 J_1^2 c_2^2 + 2 J_3 J_1^2 c_2^2 J_2 + J_2^2 c_2^2 J_1^2 \right)^{1/2} \end{aligned}$$

Die gleiche Lösung erhält man, wenn im Befehl **solve** als erstes Argument der unveränderte Ausdruck  $z[5][2]$  und als zweites  $\omega^2$  angegeben wird, allerdings mit der Maple-Warnung, dass diese Notierungsform nicht zu empfehlen ist:

Warning, solving for expressions other than names or functions is not recommended.

Die Eigenkreisfrequenzen erhalten wir nun durch Berechnung der Quadratwurzeln aller Elemente der Lösungsfolge  $EWq$ . Zu diesem Zweck wird  $EWq$  als Liste notiert und der Befehl **map** benutzt. Dieser bewirkt, dass die Funktion **sqr**t auf alle Elemente der Liste  $[EWq]$  angewendet wird. Das Ergebnis ist dann die Liste  $EW$ .

```
> EW:= map(sqr, [EWq]) ;
```

Das Ergebnis unterscheidet sich in der Darstellung von der für  $EWq$  nur dadurch, dass an die Ausdrücke für die Quadrate der Eigenkreisfrequenzen der Exponent  $\frac{1}{2}$  angehängt ist.

*Numerische Auswertung*

Für das im Bild 5.11 dargestellte Schwingungssystem seien folgende Parameterwerte bekannt:

```
> param3:= J[1]=1, J[2]=2, J[3]=1, c[1]=100, c[2]=200:
```

Mit dem Parametersatz *param3* liefert die Auswertung von *EW* folgende Werte für die Eigenkreisfrequenzen:

```
> EW3:= evalf(subs(param3, EW));
```

```
EW3:= [0.000, 18.113, 11.042]
```

Das im Bild 5.11 dargestellte System mit drei Drehmassen hat demnach drei Eigenfrequenzen, wenn  $\omega_1 = 0$  mitgezählt wird [HoDre94, S. 191]. Im Folgenden interessieren jedoch nur die beiden Werte, die größer als Null sind. Der kleinere davon ist die Grundschiwingung mit der Kreisfrequenz  $\omega = 11,042 \text{ s}^{-1}$ , der größere die Oberschiwingung mit  $\omega = 18,113 \text{ s}^{-1}$ . Zur Vereinfachung zukünftiger Operationen mit Grund- und Oberschiwingung werden die betreffenden Werte separaten Variablen zugewiesen.

```
> GS:= EW3[3]: OS:= EW3[2]:
```

### Ermittlung der Amplituden $A_i$ und der Amplitudenverhältnisse der Drehmassen

Zur Ermittlung der Amplituden wird auf die 1. Komponenten der oben berechneten Zustandsvektoren zugegriffen.

Amplitude von Drehmasse  $J_1$ :

```
> A1:= z[1][1];
```

$$A1 := \phi_0$$

Amplitude von Drehmasse  $J_3$ :

```
> A2:= z[3][1];
```

$$A2 := \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1}$$

Amplitude von Drehmasse  $J_5$

```
> A3:= z[5][1];
```

$$A3 := \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} - \frac{J_2 \omega^2 \left( \phi_0 - \frac{J_1 \omega^2 \phi_0}{c_1} \right) + J_1 \omega^2 \phi_0}{c_2}$$

Daraus ergeben sich die symbolischen Ausdrücke für die Amplitudenverhältnisse  $v_i$ . Alle Amplituden werden auf  $A_1 = \phi_0$  bezogen.

```
> v[1]:= 1:
```

```
> v[2]:= simplify(A2/A1);
```

$$v_2 := -\frac{-c_1 + J_1 \omega^2}{c_1}$$

```
> v[3]:= simplify(A3/A1);
```

$$v_3 := \frac{c_2 c_1 - J_1 \omega^2 c_2 - \omega^2 J_2 c_1 + \omega^4 J_2 J_1 - \omega^2 J_1 c_1}{c_2 c_1}$$

```
> v[3] := collect(v[3], omega);
```

$$v_3 := 1 + \frac{J_2 J_1 \omega^4}{c_2 c_1} + \frac{(-J_1 c_2 - J_2 c_1 - J_1 c_1) \omega^2}{c_2 c_1}$$

### Darstellung der Eigenschwingformen

Eigenschwingformen (Eigenformen) beschreiben die Form einer Schwingung bei einer bestimmten Eigenfrequenz. Sie ergeben sich durch Einsetzen der betreffenden Frequenz in die Ausdrücke für die Amplitudenverhältnisse. Die Eigenschwingformen sind bei der Bewertung des Einflusses von Parameteränderungen auf die Eigenfrequenzen und für die Analyse der Anregbarkeit erzwungener Schwingungen hilfreich [HoDre94].

Eigenschwingform der Grundschiwingung:

```
> V1 := evalf(subs([param3, omega=GS], [v[1], v[2], v[3]]));
V1 := [1.000, -0.219, -0.562]
```

Eigenschwingform der Oberschiwingung:


```
> V2 := evalf(subs([param3, omega=OS], [v[1], v[2], v[3]]));
V2 := [1.000, -2.281, 3.562]
```

Positionen der Drehmassen:

```
> x := [0, 1, 3]:
```

Graphische Darstellung Eigenschwingformen der Grundschiwingung (schwarz) und der Oberschiwingung (blau):

```
> with(plots): multiple(plot, [x, V1, color=black], [x, V2, color=blue],
    title="Eigenschwingformen \n von Grund- und Oberschiwingung",
    titlefont=[HELVETICA, 12]);
```



Die Nulldurchgänge der Kurven geben die Stellen der verbindenden Welle an, an denen keine Verdrehung durch die betreffende Schwingung auftritt (Schwingungsknoten). Für die Grundschiwingung ist die Differenz der Amplitudenverhältnisse zwischen Drehmasse 1 und 2 wesentlich größer als zwischen Drehmasse 2 und 3 (Bild 5.11). Die Federsteifigkeit  $c_1$  hat demnach auf die Frequenz der Grundschiwingung einen stärkeren Einfluss als  $c_2$ .



### Ermittlung der Momente

Die Momente in den Wellenabschnitten sind die 2. Größen in den Zustandsvektoren  $\mathbf{z}_k$ .

Moment zwischen den Drehmassen 1 und 2:

```
> M1 := z[1][2];
```

$$M1 := J_1 \omega^2 \phi_0$$

Moment zwischen den Drehmassen 2 und 3:

```
> M2 := collect(z[3][2], omega);
```

$$M2 := -\frac{J_2 J_1 \phi_0 \omega^4}{c_1} + (J_2 \phi_0 + J_1 \phi_0) \omega^2$$

Momentenamplituden für die Grundschiwingung:

```
> evalf(subs([param3, omega=GS], [M1, M2]));
```

$$\left[ 121.922 \phi_0, 68.466 \phi_0 \right]$$

Momentenamplituden für die Oberschwingung:

```
> evalf(subs([param3, omega=OS], [M1, M2]));
```

$$\left[ 328.078 \phi_0, -1168.466 \phi_0 \right]$$

Ist Drehwinkel  $\phi_0$  bekannt, dann kann man die exakte Größe der Drehmomente berechnen.

### Literatur:

- |           |  |
|-----------|--|
| [Göld67]  | Göldner, H.; Sähn, S.: Technische Mechanik. Dynamik. 19. Lehrbrief. VEB Verlag Technik Berlin 1967   |
| [HoDre94] | Holzweißig, F.; Dresig, H.: Lehrbuch der Maschinendynamik. 4. Aufl. Fachbuchverlag Leipzig-Köln 1994 |

Ausgleichsvorgänge in elektro-mechanischen Systemen  
mit Maple analysieren

Grundwissen für Antriebstechnik und Mechatronik

Müller, R.

2011, XII, 284 S. 69 Abb. Mit zahlreichen Beispielen und  
Maple-Plots., Softcover

ISBN: 978-3-8348-1217-9